Math 1. 2-hours-test December 8, 2019.

JE/JKL 8.12.19

Problem 1

```
> restart;with(LinearAlgebra):
```

A homogeneous system of linear equations is given by

$$x_1 + x_2 + x_3 + x_4 = 0$$
$$2\,x_1 - x_2 + 8\,x_3 - 4\,x_4 = 0$$
$$x_1 - 2\,x_2 + 7\,x_3 + a x_4 = 0$$

where a is an arbitrary real number.
The coefficient matrix is

```
> A:=a-><1,1,1,1;2,-1,8,-4;1,-2,7,a>:
> A(a);
```

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 2 & -1 & 8 & -4 \\ 1 & -2 & 7 & a \end{bmatrix} \qquad \textbf{(1.1)}$$

Question 1

For $a = 1$ the augmented matrix is

```
> T:=<A(1)|0,0,0>;
```

$$T := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 2 & -1 & 8 & -4 & 0 \\ 1 & -2 & 7 & 1 & 0 \end{bmatrix} \qquad \textbf{(1.1.1)}$$

that has the completely reduced form

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 3 & 0 & 0 \\ 0 & 1 & -2 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \end{bmatrix} \qquad \textbf{(1.1.2)}$$

Then the completely reduced system of linear equations is

$x_1 + 3\,x_3 = 0$

$x_2 - 2\,x_3 = 0$

$x_4 = 0$

If we put $x_3 = t$ we get the complete solution on standard parametric form

$(x_1, x_2, x_3, x_4) = t\,(-3, 2, 1, 0)$, $t \in \mathbb{R}$.

Question 2

```
> T:=<A(a)|0,0,0>;
```

$$T := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 2 & -1 & 8 & -4 & 0 \\ 1 & -2 & 7 & a & 0 \end{bmatrix} \quad \textbf{(1.2.1)}$$

```
> T2:=RowOperation(T,[2,1],-2);
```

$$T2 := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 0 & -3 & 6 & -6 & 0 \\ 1 & -2 & 7 & a & 0 \end{bmatrix} \quad \textbf{(1.2.2)}$$

```
> T3:=RowOperation(T2,[3,1],-1);
```

$$T3 := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 0 & -3 & 6 & -6 & 0 \\ 0 & -3 & 6 & a-1 & 0 \end{bmatrix} \quad \textbf{(1.2.3)}$$

From **T**3 we read that $\rho(\mathbf{A}(a)) = 2 \Leftrightarrow a-1 = -6 \Leftrightarrow a = -5$.
For $a = -5$ the augmented matrix is

```
> T:=<A(-5)|0,0,0>;
```

$$T := \begin{bmatrix} 1 & 1 & 1 & 1 & 0 \\ 2 & -1 & 8 & -4 & 0 \\ 1 & -2 & 7 & -5 & 0 \end{bmatrix} \quad \textbf{(1.2.4)}$$

that has the completely reduced form

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 3 & -1 & 0 \\ 0 & 1 & -2 & 2 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad \textbf{(1.2.5)}$$

Then the completely reduced system of linear equations is
$x_1 + 3\,x_3 - x_4 = 0$
$x_2 - 2\,x_3 + 2\,x_4 = 0$
if we put $x_3 = t_1$ and $x_4 = t_2$ we get the complete solution in standard parametric form
$(x_1, x_2, x_3, x_4) = t_1(-3, 2, 1, 0) + t_2(1, -2, 0, 1)$, $t_1, t_2 \in \mathbb{R}$.
For $(t_1, t_2) = (1, 0)$ we get the solution $(-3, 2, 1, 0)$ and for $(t_1, t_2) = (0, 1)$ we get the solution $(1, -2, 0, 1)$. These two solution af linearly independent since they are not proportional.
Thus for $a = -5$, $(-3, 2, 1, 0)$ and $(1, -2, 0, 1)$ are two liearly independent solutions to the system of equations.

Probem 2

```
> restart;with(LinearAlgebra):
```

A 2-dimensional vector space V has the base $v = (\mathbf{v}_1, \mathbf{v}_2)$.
$f : \mathrm{V} \rightarrow \mathbb{R}^3$ is linear and the mapping matrix ${}_e\mathbf{F}_v = [\,{}_e f(\mathbf{v}_1)\ {}_e f(\mathbf{v}_2)]$ is

```
> eFv:=<1,2;0,-1;2,0>;
```

$$eFv := \begin{bmatrix} 1 & 2 \\ 0 & -1 \\ 2 & 0 \end{bmatrix} \quad \textbf{(2.1)}$$

Question 1

The vector $\mathbf{v}_3 = 2\,\mathbf{v}_1 - 5\,\mathbf{v}_2$ has the coordinate vector ${}_v\mathbf{v}_3 = \begin{bmatrix} 2 \\ -5 \end{bmatrix}$ with respect to the base v.

${}_e f(\mathbf{v}_3) = {}_e\mathbf{F}_v\ {}_v\mathbf{v}_3$

```
> eFv.<2,-5>;
```

$$\begin{bmatrix} -8 \\ 5 \\ 4 \end{bmatrix} \quad \textbf{(2.1.1)}$$

I.e. $f(\mathbf{v}_3) = (-8, 5, 4)$.

Question 2

$f(\mathbf{v}) = (1, 2, 10) \Leftrightarrow {}_e f(\mathbf{v}) = \begin{bmatrix} 1 \\ 2 \\ 10 \end{bmatrix} \Leftrightarrow {}_e\mathbf{F}_v\ {}_v\mathbf{v} = \begin{bmatrix} 1 \\ 2 \\ 10 \end{bmatrix}$.

The augmented matrix is

```
> T:=<eFv|1,2,10>;
```

$$T := \begin{bmatrix} 1 & 2 & 1 \\ 0 & -1 & 2 \\ 2 & 0 & 10 \end{bmatrix} \quad \textbf{(2.2.1)}$$

that has the completely reduced form

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 5 \\ 0 & 1 & -2 \\ 0 & 0 & 0 \end{bmatrix} \quad \textbf{(2.2.2)}$$

From this we read that ${}_v\mathbf{v} = \begin{bmatrix} 5 \\ -2 \end{bmatrix}$ and thus the solution vector is $\mathbf{v} = 5\,\mathbf{v}_1 - 2\,\mathbf{v}_2$.

Question 3

$f(\mathrm{V}) = \mathrm{span}\{f(\mathbf{v}_1), f(\mathbf{v}_2)\} = \mathrm{span}\{(1, 0, 2), (2, -1, 0)\}$.

$\dim(f(\mathrm{V})) = \rho({}_e\mathbf{F}_v) = 2$. (Follows e.g. from trap(**T**) in Question 2.)

Thus the image space (the range) $f(\mathrm{V}) = \mathrm{span}\{(1, 0, 2), (2, -1, 0)\}$ is a plane in $\mathbb{R}^3$ through the origin.

$\dim(\ker(f)) = \dim(\mathrm{V}) - \dim(f(\mathrm{V})) = 0$.

Question 4

$\mathbf{u} = (u_1, u_2, u_3) \in f(\mathrm{V}) \Leftrightarrow f(\mathbf{x}) = \mathbf{u}$ has a solution $\mathbf{x} = x_1\mathbf{v}_1 + x_2\mathbf{v}_2 \in \mathrm{V} \Leftrightarrow {}_e\mathbf{F}_v \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} =$

$\begin{bmatrix} u_1 \\ u_2 \\ u_3 \end{bmatrix}$ has a solution $(x_1, x_2) \in \mathbb{R}^2$.

If $\mathbf{u} = \mathbf{e}_1 = (1, 0, 0)$ then the matrix equation has the augmented matrix

```
> T:=<eFv|1,0,0>;
```

$$T := \begin{bmatrix} 1 & 2 & 1 \\ 0 & -1 & 0 \\ 2 & 0 & 0 \end{bmatrix} \qquad \textbf{(2.4.1)}$$

that has the completely reduced form

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \textbf{(2.4.2)}$$

Since $\rho({}_e\mathbf{F}_v) = 2 < 3 = \rho(\mathbf{T})$ the matrix equation has no solutions which means that $\mathbf{e}_1 = (1, 0, 0)$ does not belong to the image space $f(\mathrm{V})$. (So, the vector$(1, 0, 0)$ does not lie in the plane $f(\mathrm{V})$ through the origin spanned by the vectors $(1, 0, 2)$ and $(2, -1, 0)$.)

Problem 3

```
> restart;with(LinearAlgebra):
```

$\mathbb{R}^3$ equipped with usual dot product.

```
> v1:=<1,-1,1>;
```

$$v1 := \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix} \qquad \textbf{(3.1)}$$

```
> v2:=<1,0,-1>;
```

$$v2 := \begin{bmatrix} 1 \\ 0 \\ -1 \end{bmatrix} \qquad \textbf{(3.2)}$$

```
> v3:=<1,1,0>;
```

$$v3 := \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix} \qquad \textbf{(3.3)}$$

About a real 3×3-matrix $\mathbf{A}$ it is given that it has the eigenspaces $\mathrm{E}_6 = \mathrm{span}\{\mathbf{v}_1\}$ and $\mathrm{E}_{-3} = \mathrm{span}\{\mathbf{v}_2, \mathbf{v}_3\}$.

Question 1

```
> V:=<v1|v2|v3>;
```

$$V := \begin{bmatrix} 1 & 1 & 1 \\ -1 & 0 & 1 \\ 1 & -1 & 0 \end{bmatrix} \quad \textbf{(3.1.1)}$$

```
> Determinant(V);
```

$$3 \quad \textbf{(3.1.2)}$$

Thus the columns in **V** are three linearly independent eigenvectors for **A**.
If we put

```
> Lambda:=DiagonalMatrix([6,-3,-3]);
```

$$\Lambda := \begin{bmatrix} 6 & 0 & 0 \\ 0 & -3 & 0 \\ 0 & 0 & -3 \end{bmatrix} \quad \textbf{(3.1.3)}$$

then $\mathbf{\Lambda} = \mathbf{V}^{-1}\,\mathbf{A}\,\mathbf{V}$.

Question 2

$t_1\mathbf{v}_1$, where $t_1 \in \mathbb{R}$, is an arbitry vector in E_6 and $t_2\mathbf{v}_2 + t_3\mathbf{v}_3$, where $(t_2, t_3) \in \mathbb{R}^2$, is an arbitrary vector in E_{-3}.
Since

```
> simplify(DotProduct(t1*v1,t2*v2+t3*v3));
```

$$0 \quad \textbf{(3.2.1)}$$

then every vector in E_6 is orthogonal to every vector in E_{-3}. I.e. the two eigenspaces E_6 and E_{-3} are orthogonal.

Question 3

```
> q1:=v1/Norm(v1,2);
```

$$q1 := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ -\frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix} \quad \textbf{(3.3.1)}$$

is an orthonormal basis for E_6.

```
> q2:=v2/Norm(v2,2);
```

$$q2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ 0 \\ -\frac{\sqrt{2}}{2} \end{bmatrix} \quad \textbf{(3.3.2)}$$

and

```
> q3:=CrossProduct(q1,q2);
```

$$q3 := \begin{bmatrix} \frac{\sqrt{3}\sqrt{2}}{6} \\ \frac{\sqrt{3}\sqrt{2}}{3} \\ \frac{\sqrt{3}\sqrt{2}}{6} \end{bmatrix} \quad \textbf{(3.3.3)}$$

constitute an orthonormal basis for E_{-3}.

Then (**q**1, **q**2, **q**3) is an orthonormal basis for $\mathbb{R}^3$ – equipped with the ordinary crossproduct – consisting of eigenvectors for **A**.

If we put

```
> Q:=<q1|q2|q3>;
```

$$Q := \begin{bmatrix} \frac{\sqrt{3}}{3} & \frac{\sqrt{2}}{2} & \frac{\sqrt{3}\sqrt{2}}{6} \\ -\frac{\sqrt{3}}{3} & 0 & \frac{\sqrt{3}\sqrt{2}}{3} \\ \frac{\sqrt{3}}{3} & -\frac{\sqrt{2}}{2} & \frac{\sqrt{3}\sqrt{2}}{6} \end{bmatrix} \quad \textbf{(3.3.4)}$$

then **Q** is positive orthogonal

```
> Transpose(Q).Q;Determinant(Q);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$1 \quad \textbf{(3.3.5)}$$

and

$$\mathbf{Q}^{-1}\,\mathbf{A}\,\mathbf{Q} = \mathbf{\Lambda} \Leftrightarrow \mathbf{A} = \mathbf{Q}\,\mathbf{\Lambda}\,\mathbf{Q}^{-1} = \mathbf{Q}\,\mathbf{\Lambda}\,\mathbf{Q}^{\mathrm{T}},$$

where $\mathbf{\Lambda}$ is the diagonalmatrix stated in question 1.

The unknown matrix is then

```
> A:=Q.Lambda.Transpose(Q);
```

$$A := \begin{bmatrix} 0 & -3 & 3 \\ -3 & 0 & -3 \\ 3 & -3 & 0 \end{bmatrix} \quad \textbf{(3.3.6)}$$

Probem 4

```
> restart;with(LinearAlgebra):with(plots):
```

$$\begin{bmatrix} x_1'(t) \\ x_2'(t) \end{bmatrix} = \mathbf{A} \begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix}, t \in \mathbb{R}$$

where **A** is a reel 2×2-matrix.

It is given that the complete complex solution to the system is given by:

$$\begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix} = c1\, e^{(2+6\,i)t} \begin{bmatrix} i \\ 1 \end{bmatrix} + c2\, e^{(2-6\,i)t} \begin{bmatrix} -i \\ 1 \end{bmatrix}, t \in \mathbb{R} \;,\; c1, c2 \in \mathbb{C}.$$

Written in Maple

```
> x:=unapply(c1*exp((2+6*I)*t)*<I,1>+c2*exp((2-6*I)*t)*<-I,1>,t):
> x(t);
```

$$\begin{bmatrix} I\, c1\, e^{(2+6\,I)\,t} - I\, c2\, e^{(2-6\,I)\,t} \\ c1\, e^{(2+6\,I)\,t} + c2\, e^{(2-6\,I)\,t} \end{bmatrix} \tag{4.1}$$

Question 1

From the given complete complex solution we read that all eigenvalues for system matrix **A** are $2+6\,i$ and $2-6\,i$ with the corresponding eigenvectorspaces

$$E_{2+6\,i} = \text{span}\left\{ \begin{bmatrix} i \\ 1 \end{bmatrix} \right\} \text{ and } E_{2-6\,i} = \text{span}\left\{ \begin{bmatrix} -i \\ 1 \end{bmatrix} \right\}.$$

Question 2

The particular solution, where

```
> c1:=1/2;
```

$$c1 := \frac{1}{2} \tag{4.2.1}$$

and

```
> c2:=1/2;
```

$$c2 := \frac{1}{2} \tag{4.2.2}$$

is

```
> x(t);
```

$$\begin{bmatrix} \frac{I\, e^{(2+6\,I)\,t}}{2} - \frac{I\, e^{(2-6\,I)\,t}}{2} \\ \frac{e^{(2+6\,I)\,t}}{2} + \frac{e^{(2-6\,I)\,t}}{2} \end{bmatrix} \tag{4.2.3}$$

Since

```
> 'x(0)'=x(0);
```

$$x(0) = \begin{bmatrix} 0 \\ 1 \end{bmatrix} \quad \textbf{(4.2.4)}$$

this particular solution fulfills the stated initial condition.

Question 3

The particular solution form the prededing question expressed in terms of real functions is

```
> xr:=t->evalc(x(t)):
> xr(t);
```

$$\begin{bmatrix} -e^{2t}\sin(6t) \\ e^{2t}\cos(6t) \end{bmatrix} \quad \textbf{(4.3.1)}$$

The trajectory that the point $(x_1(t), x_2(t))$ traverses in time $t \in [0, 1]$ is

```
> P1:=plot([xr(t)[1],xr(t)[2],t=0..1],scaling=constrained,
  tickmarks=[6,11],color=blue):
> tekst:=textplot([[0.4,1,"P"],[-1.4,-0.3,"Q"],[0.4,-2.55,"R"]
  ]):
> display(P1,tekst);
```



Since $x_2(t) = e^{2t}\cos(6t) = 0 \Leftrightarrow 6t = \frac{\pi}{2} + p\pi, p \in \mathbb{Z} \Leftrightarrow t = \frac{\pi}{12} + p\frac{\pi}{6}, p \in \mathbb{Z}$, the curve intersects the first axis for the first time when $t = \frac{\pi}{12}$.

The point Q on the first axis, that the curve passes at $t = \frac{\pi}{12}$ is

```
> evalf(xr(Pi/12));
```

$$\begin{bmatrix} -1.688091795 \\ 0. \end{bmatrix} \tag{4.3.2}$$

which looks reasonable as compared with the figure.

Since $x_1(t) = -e^{2t}\sin(6t) = 0 \Leftrightarrow 6t = p\pi,\ p \in \mathbb{Z} \Leftrightarrow t = p\frac{\pi}{6},\ p \in \mathbb{Z}$, the curve intersects the second axis for the first time when $t = \frac{\pi}{6}$.

The point R on the first axis, that the curve passes at $t = \frac{\pi}{6}$ is

```
> evalf(xr(Pi/6));
```

$$\begin{bmatrix} 0. \\ -2.849653908 \end{bmatrix} \tag{4.3.3}$$

which looks reasonable as compared with the figure.

One could also directly use the complex solution

```
> P2:=plot([x(t)[1],x(t)[2],t=0..1],scaling=constrained,
  tickmarks=[6,11],color=blue):
> tekst:=textplot([[0.4,1,"P"],[-1.4,-0.3,"Q"],[0.4,-2.55,"R"]
  ]):
> display(P2,tekst);
```



Intersection with the first axis

```
> solve(x(t)[2]=0,t,allsolutions);
```

$$-\frac{1}{6}\pi\_Z1{\sim}-\frac{1}{12}\pi \quad \textbf{(4.3.4)}$$

First positive solution is again $t=\frac{\pi}{12}$.

Intersection with the second axis

```
> solve(x(t)[1]=0,t,allsolutions);
```

$$-\frac{\pi\_Z2{\sim}}{6} \quad \textbf{(4.3.5)}$$

First positive solution is again $t=\frac{\pi}{6}$.